
第3章 関数型言語Haskellとは

Haskellは と呼ばれ、ラムダ計算を基本としながらも実際の使用に便利な機能を追
加したプログラミング言語である。Haskellと前章で紹介した（型なし）ラムダ計算との主な違いは

1. 式は されている。

2. を定義することができる。

3. を使用することができる。

4. グラフ簡約によって実行される。

5. 中置記法を使用することができる。

などの点である。また、他のプログラミング言語と比べた時の特徴は、以下のようになる。

1. を持ち、プログラマがメモリの管理に煩わされることがない。

2. 関数を他の関数の引数としたり、関数を生成して他の関数の戻り値としたり、関数を値として
使うことができる（ ）。

3. を許すことにより、汎用性のある関数を定義することができる。

4. により、（ほとんどの場合）プログラム中に型を書く必要はない。 など、型
システムが発達している。

5. 式に はない。入出力や参照の書換えなどの効果も値として表現される。このため、プ
ログラムの同値性などの性質に関する考察が、より容易になる。

6. を採用し、グラフ簡約によって実行される。

一般に関数型言語は記号処理に適している。純関数型言語を Cや Javaなどと同じように実世界の
プログラミングに利用することも可能である。しかし、ここでは主にプログラミング言語の意味を説
明するための超言語として紹介することにする。

3.1 Haskell処理形の入手とインストール

Haskellの処理形としてここでは GHC (Glasgow Haskell Compiler1)を利用することにする。GHCは
Haskellの処理系の事実上の標準（デファクトスタンダード）である。GHCは、http://www.haskell.

1Guarded Horn Clausesという論理プログラミング言語と同じ頭文字だが、何の関係もない。

org/ghc/からダウンロードすることができる。Windowsの場合、インストールは入手したファイル
をダブルクリックするだけである。
また Linux用の RPMファイルなども上記のホームページから入手することができる。

3.2 GHCiのコマンド

GHCは、多くのプログラムの集合体であり、gccや javacのように実行可能形式を出力するバッ
チコンパイラ（ghc）もその中に含まれている。
ここでは、その中で対話型の処理系である GHCi(ghci)の使用法を説明する。GHCiを起動すると、

Prelude>

のようなプロンプトが表示される。このプロンプトからコマンドを入力することによってファイルか
らプログラムをロードし、式を評価することができる。

GHCiのコマンドには次のようなものがある。

コマンド 省略形 意味
:load file :l fileをロードする。
:also file :a fileを追加ロードする。
:reload :r 以前にロードしたファイルをリロードする。
:type expr :t exprの型を表示する。
:cd directory :c ディレクトリを変更する。
:! command commandをコマンドプロンプトに渡して実行する。
:? ヘルプを表示する。
:quit :q GHCiを終了する。

また、単に式を入力すると、その式を評価してその結果を出力する。

Prelude> 1+2
3

3のように斜体になっているところがシステムの出力で、それ以外がユーザの入力である。
Haskellのプログラムファイルには通常 という拡張子をつける。

3.3 Haskellのプログラムの基本

Haskellの仕様書は http://www.haskell.org/から入手することができる。以下では、Haskellの
仕様の基本的なところを紹介する。

変数の宣言 Haskellでは他のプログラミング言語同様、変数を宣言して良く使う式に名前をつける
ことができる。ただし、C言語のような命令型言語と異なり、変数は一度宣言するとその値を変える
ことはできない（代入はできない）。
変数の宣言は次の形で行なう。

変数名 = 式

複数の変数をまとめて宣言する時は次の形式になる。

{
変数名 1 = 式 1;
変数名 2 = 式 2;
. . . ;
変数名 n = 式 n

}

つまり、前後を “{”と “}”（ブレース）で囲み、“; ”（セミコロン）で区切る。ただし、ブレースとセ
ミコロンは多くの場合省略でき、以下の例でも原則として省略する。省略の条件は付録で説明する。
変数名には C言語と同じようにアルファベット、数字、“_”（アンダースコア）が使えるほか、 “’”

（アポストロフィ）も使うことができる。ただし、通常の変数名は （またはアンダースコア）
から始まる必要がある。大文字から始まる名前は、後で紹介する構成子名に用いる。

プログラム Haskellのプログラムはモジュールの集合で、1つのモジュールは基本的には複数の変数
の宣言の前に

module モジュール名 where

というヘッダ部分をつけた形式である。つまり、以下のようなかたちである。

module モジュール名 where {
変数名 1 = 式 1;
変数名 2 = 式 2;
. . . ;
変数名 n = 式 n

}

ただし変数の宣言の他に、import宣言や型の宣言、型クラス関係の宣言などを書くことができる。こ
れらについては後述する。通常、1つのファイルに 1つのモジュールを記述する。
「moduleモジュール名 where」の部分を省略すると Mainという名前のモジュールの定義と解釈さ
れる。ブレースやセミコロンも多くの場合省略されるので、もっとも簡単な場合、Haskellのプログ
ラムは単に次のような形式をしていることになる。

変数名 1 = 式 1
変数名 2 = 式 2
. . .
変数名 n = 式 n

ラムダ式 Haskellでは、関数を表すのに当然ラムダ式に似た記法を用いることができる。ただし、
“λ” の代わりに “ ”（バックスラッシュまたは “Y=”）、“ .”（ピリオド）の代わりに “ ” を用いる。

ラムダ計算 Haskell

λx.x

λxy.y

関数の適用はラムダ計算と同様、関数と実引数を並べて書くだけである。通常の数学の記法や C言
語などのように引数に括弧をつける必要はない。

c0 = \ f x -> x
c1 = \ f x -> f x
c2 = \ f x -> f (f x)
true = \ t f -> t
false = \ t f -> f

実は、このようにラムダ式に名前をつける時は、仮引数を “=” の左辺に並べて、
c0 f x = x
c1 f x = f x
c2 f x = f (f x)
true t f = t
false t f = f

のように書くことができる。（むしろ、このように書く方が普通である。）

コメント Haskellのコメントには 2つの形がある。 という形と

という形である。

3.4 組み込みのデータ型と演算子

Haskellでは真偽値（Bool）、整数（Int, Integer—Intは固定（有限）精度, Integerは無限精度）、
浮動小数点数（Float, Double）、文字（Char）などは組み込みのデータ型として用意されている。
Bool型のリテラルは と であり、Integer, Float, Double, Char型などのリテラルの記
法は C言語とほぼ同様である。
これらのデータ型に対しては組み込みの関数や演算子がいくつか用意されている。Lispの場合と異
なり、算術演算子の “+”, “ -”, “ *” などは、1+2*3のように通常の中置記法で用いる。
if～then～else式も組み込みで用意されている。次の形で用いる。
if 式 1 then 式 2 else 式 3

式 1は でなければならず、式 2と式 3は同じ型でなければならない。式 1が Trueの時は式 2、
Falseの時は式 3として評価される。なお、if～then～else式は、Haskellでは後で説明するように
特殊な評価順を必要とする特殊形式ではない。
次の例は階乗（ factorial）の関数の Haskellでの宣言である。
fact n = if n==0 then 1 else n*fact(n-1)

関数適用は他のどんな中置記法の演算子よりも 。そのため、fact(n-1)は fact

n-1と書くことはできない。後者は の意味になってしまう。逆に n*(fact(n-1))の
外側の括弧は必要ない。
なお、この例のように変数は することが可能である。再帰的定義に特別な文法を使
用する必要はない。

問 3.4.1 fact 100を計算してみよ。

リスト リスト型も組み込みのデータ型として用意されている。リストとは簡単に言えばデータの並
びである。リストは伝統的に Schemeなどの Lisp系の言語が得意とするデータ型であり、Haskellで
も豊富なライブラリ関数が用意されている。
リストは、空リスト とコンス (cons)と呼ばれる演算子 “ ” から構成される。

• 空リスト（[]）は文字通り空のリストである。

• コンス（:）は右オペランドとして渡されるリストの先頭に左オペランドとして渡される要素を
付け加えたリストを返す演算子である。

また、“:” は である。つまり、1:2:[]は 1:(2:[])のことを表す。
リストのリテラルの記法として、要素を “,”（コンマ）で区切って並べ、 “[” と “]” で囲む記法も

用意されている。例えば [1,2,3,4]は、 のことである。
リスト型はパラメトリックな型である。つまり、リストの要素の型をパラメータとする。要素の型

が Integer型の場合、そのリストの型は 型、要素の型が Double型の場合リストの型

は 型と書き表される。Haskellの文字列（String）型は実は文字のリスト型として表さ
れている。つまり、

type String = [Char]

のように定義されている。ここで、

type 型名 = 型

は型の別名（ type alias）を宣言する形式である。上の例の場合 Stringという型名が、[Char]という
型の別名となる。型名は大文字から始まる識別子でなければならない。

box-pointer記法 リストを、箱と矢印を用いた図で表すことがある。例えば、次のように構成され
たリストの場合、

xs0 = []
xs1 = 1:xs0
xs2 = 2:xs1
xs3 = 3:xs2

次の図のようになる。

3 2 1

xs3 xs2 xs1

1つの :を 2個の正方形の箱がつながった箱（コンスセル）への矢印で表す。箱の中身は、1, 2, 3

などの数、他の箱への矢印などである。（箱の中に他の箱がまるごと入ることはない。）空リストは斜
線を引いて表す。
また、[[1,2],[3,4]]というリストのリストは

1 2 3 4

という箱矢印記法で表される。

問 3.4.2次のリストを box-pointer記法で書け。

1. [2,3,5,7,11]

2. [0]

3. [[1],[2,3,4],[]]

（参考）
リストを C言語の構造体として定義すると次のようになる。

struct _list {
int car;
struct _list* cdr;

};

typedef struct _list* list;

_listという構造体は、carと cdrという 2つのメンバを持つ。このうち cdrは現在定義されてい
る型へのポインタ型である。listという型は、typedef宣言によって、struct _list*という型の
別名として定義される。（箱矢印記法で矢印になっているところが、C言語ではポインタとして表さ
れる。）
また、空リストは C言語では通常定数 NULLで表される。
malloc関数が使われていることからわかるように、C言語でリストを扱う場合は、いつ freeをす

るかを気を付けなければいけない。Haskellや他の関数型言語では明示的に freeしなくても、いらな
くなったメモリ領域は自動的に回収されることになっている。

組 組（ tuple）も組み込みのデータ型として用意されている。組は要素を “,”（コンマ）で区切って
並べ、 “(”と “)”で囲んで表す。組はリストの場合と異なり、要素の型が同一である必要はない。(1,
’a’)という式の型は と表記される。また、(2, ’b’, [3])という式の型は

と表記される2。

関数型 関数の型は “->”という記号を使って、Integer -> Charのように表記される。これは、引数
の型が Integerで戻り値の型が Charの関数の型である。“->”は である。つまり、Bool ->

Bool -> Boolという型は と解釈される。多引数関数をカリー化
して定義する場合、このように約束しておく方が便利である。

2実際の Haskellでは、型クラス（ type class）というものが関係するため、これらの式はもう少し複雑な型を持つ。ここ
では型クラスの説明は避けて、実際よりも単純化した型を紹介する。

多相型 Integerや Charのように型定数の名前は必ず大文字から始まる。一方、aや bのように小文
字で始まる識別子が型の中に現れる場合、これらは である。これらの型変数は使用する時に、
より具体的な型に置き換えることができる。例えば、[a] -> [b] -> [(a, b)]という型を持つ関数
は、[Char] -> [Integer] -> [(Char, Integer)]という型を持つ関数として使用しても良いし、
[String] -> [Integer -> Integer] -> [(String, Integer -> Integer)]として使用しても
良い。
このように型変数を含む型を という。Haskellは とともに多相型を許す代表的なプロ

グラミング言語である。
なお、変数の型をプログラム中に明示したい場合 “::” という記号を使って、

変数名 :: 型

と書く。例えば factの型を明示しておきたい場合は、
fact :: Integer -> Integer
fact n = if n==0 then 1 else n*fact(n-1)

のように書く。ただし通常は、変数の型はプログラマが明示しなくても、Haskell処理系が推論して
くれる。この仕組みを という。

3.5 パターンマッチング

Haskellでは、関数の仮引数の部分に というものを書いて、引数の形に応じて場合分け
を行なう事が可能である3。パターンとは、大雑把に言って変数と定数、構成子からのみ生成される
式である。

-- Preludeの lengthとほぼ同じ
myLength :: [a] -> Integer
myLength [] = 0
myLength (x:xs) = 1 + myLength xs

この例の場合、myLengthの引数が空リスト（[]）ならば 1行目が選択される。一方、1つ以上の要
素を持つリストならば 2行目が選択され、リストの先頭要素が変数 xに束縛され、残りのリストが変
数 xsに束縛される。（なお、myLengthとほとんど同じ関数 length :: [a] -> Intが標準ライブ
ラリに用意されている。）
パターンマッチングで、右辺で使わない部分には “_”（アンダースコア）を使って変数に束縛せず、
無視することができる。例えば myLengthの場合、xは右辺で使用していないので、

myLength (_:xs) = 1 + myLength xs

と書くことができる。
case～of式もパターンマッチングを行なう。case～of式は次の形で用いる。（やはり、ブレース

とセミコロンは通常省略する。）

case 式 0 of {
パターン 1 -> 式 1;
パターン 2 -> 式 2;

3一般に関数型言語はデータの種類は増えずに処理（関数）が増えるような場合が得意、オブジェクト指向言語は、デー
タ（クラス）の種類が増えて、処理（メソッド）が増えないような場合が得意である。

. . .
パターン n -> 式 n

}

式 0を評価して、上から順にパターンと照合し、パターン 1にマッチするならば式 1が、パターン m

にマッチするならば式 mがそれぞれ評価される。
if 式 1 then 式 2 else 式 3という式も次の case～of式の略記法と解釈することが可能である。

case 式 1 of { True -> 式 2; False -> 式 3 }

この他に、let式（後述）や λ式など変数を束縛するところでもパターンを書くこともできる。例
えば、

\ (x:xs) -> x

などである。（この関数は headという名前で標準ライブラリに用意されている。）この場合は、パター
ンは一種類のみで場合分けはできず、パターンにマッチしない引数が与えられればエラーとなる。

問 3.5.1リスト中の数の和を求める関数 mySumを定義せよ。

問 3.5.2真偽値のリスト [Bool]を 2進数と見なして、対応する整数を計算する関数 fromBin ::

[Bool] -> Integerを定義せよ。例えば、fromBin [True, True]は 3, fromBin [True, False,

True, False]は 10になる。
ヒント : 引数の数を一つ増やした補助関数が必要になる。

問 3.5.3実数のリスト xsと実数から実数への関数 fを受け取り、リストの各要素に fを適用した結
果の和を計算する関数 sumf :: [Double] -> (Double -> Double) -> Doubleを定義せよ。

3.6 帰納法による証明

プログラミング言語の意味をラムダ計算や関数型言語で表現することの利点は、プログラムの同値
性（等価性）などの議論が容易になるというところにある。ここでは、そのような議論の例として、
2つのリストに関する関数が等価であることの証明を取り上げる。このような証明は帰納法と併用す
ることが多い。以下の例も帰納法を使用している。
リストを反転する関数 reverse:
-- appendは Preludeの (++)演算子と同じ
append :: [a] -> [a] -> [a]
append [] ys = ys
append (x:xs) ys = x : (append xs ys)

-- reverseは Preludeに定義済み
reverse :: [a] -> [a]
reverse [] = []
reverse (x:xs) = append (reverse xs) [x]

は上の定義では、引数の に比例する時間がかかるために効率が悪い。

そこで次のような定義を考える。
-- shunt は revの補助関数
shunt :: [a] -> [a] -> [a]
shunt ys [] = ys
shunt ys (x:xs) = shunt (x:ys) xs

rev :: [a] -> [a]
rev xs = shunt [] xs

この revという関数は、引数の に比例する時間でリストを反転できるの
で効率が良い。
revと reverseが等価であること –正確に言うと、すべての有限リスト xsに対して

rev xs = reverse xs

が成り立つことを証明できる。
そのためには、次のような補助定理を証明すれば良い。

shunt ys xs = append (reverse xs) ys

これは、 で証明することができる。
証明:
xs = []のとき:

xs = z:zsのとき:

問 3.6.1すべての有限リスト xsについて、

append xs (append ys zs) = append (append xs ys) zs

が成り立つことを、xsに関する帰納法で証明せよ。

3.7 有用なリストの高階関数

次のようなリストに対する高階関数がよく利用される。これらは Prelude（標準ライブラリ）に定
義済みである。

map :: (a -> b) -> [a] -> [b]
map f [] = []
map f (x:xs) = f x : map f xs

zipWith :: (a -> b -> c) -> [a] -> [b] -> [c]
zipWith f (x:xs) (y:ys) = f x y : zipWith xs ys
zipWith f _ _ = []

filter :: (a -> Bool) -> [a] -> [a]
filter p [] = []
filter p (x:xs) = if p x then x : filter p xs else filter p xs

iterate :: (a -> a) -> a -> [a]
iterate f x = x : iterate f (f x)

foldr :: (a -> b -> b) -> b -> [a] -> b
foldr f x [] = x
foldr f x (y:ys) = f y (foldr f x ys)

foldl :: (a -> b -> a) -> a -> [b] -> a
foldl f x [] = x
foldl f x (y:ys) = foldl (f x y) ys

3.8 その他の便利な記法

関数の中置記法化 Haskellの関数は通常は前置記法で用いるが、識別子を “ ”（バッククォート、
クォート（ “’”）ではないことに注意）で囲むことによって、中置記法で書くことができる。これは小
さなことに見えるが実は意外に便利である。例えば、次のように Preludeで定義された zipの場合、

zip :: [a] -> [b] -> [(a,b)]
zip (a:as) (b:bs) = (a,b) : zip as bs
zip _ _ = []

通常は zip [1, 2] [3, 4]のように書くが、これを [1, 2] ‘zip‘ [3, 4]と書いても良い。
中置記法で用いる演算子に対して、infixl, infixr, infixなどのキーワードを使って、優先順位

と結合性を定めることができる。たとえば、Prelude（Haskellにはじめから読み込まれる標準ライブ
ラリ）では次のように宣言されている。

infixr 9 .
infixl 9 !!
infixr 8 ˆ, ˆˆ, **
infixl 7 *, /, ‘quot‘, ‘rem‘, ‘div‘, ‘mod‘, :%, %
infixl 6 +, -
infixr 5 :
infixr 5 ++
infix 4 ==, /=, <, <=, >=, >, ‘elem‘, ‘notElem‘
infixr 3 &&
infixr 2 ||
infixl 1 >>, >>=
infixr 1 =<<
infixr 0 $, $!, ‘seq‘

infixlは 、infixrは を表す。ただの infixはどちらでもないこと（非結合—例
えば 1 < x < 2は構文エラー！）を表す。また、2列目の数字が大きいほど、優先順位が高い。例
えば*は +よりも結合力が強い。
バッククォートと逆に本来中置記法で使用される演算子を “(” と “)”でくくって、ふつうの前置記
法で用いることができる。例えば1+2を(+) 1 2と書くことができる。あるいは関数の引数として渡
すこともできる。

局所的定義 letというキーワードを用いて、局所的な変数を定義することができる。

let （複数の）変数の定義 in 式

という形で用いる。
pow4 x = let y = x*x in y*y
-- headは Preludeに定義済み
head ys = let (x:xs) = ys in x

などである。この例では 4乗する関数（pow4）を定義しているが、y*yの部分が変数 yの有効範囲
（ ）に属している。実は、さらに “変数の定義” の右辺の部分（この例では、 の部分）
もスコープに属している。これは次の例でわかる。

-- repeatは Preludeに定義済み
repeat :: a -> [a]
repeat x = let xs = x:xs in xs

この関数は要素 xの無限リストを生成する。

Prelude> repeat 1
[1,1,1,1,1,1,1,1,1,1,1,1,1,1,1,1,1,1,1,1,1,1, . . . 4

（このような定義は、Haskellでは “止まらない・役に立たない式”ではなく、意味のある式となる。こ
のことは、あとで Haskellの評価戦略を紹介する時に説明する。）

問 3.8.1リストを昇ベキの順に表された多項式と見なし、多項式の値を計算する関数
evalPoly :: [Double] -> Double -> Doubleを定義せよ。例えば、[1, 2, 3, 4]というリスト
は 1 + 2x + 3x2 + 4x3という多項式と見なし、evalPoly [1, 2, 3, 4] 10の値は 4321になる。

4Ctrl-cで中断する。

問 3.8.2リストを集合だと見なして、そのベキ集合（部分集合の集合）を返す関数 powersetを定義
せよ。
例えば powerset [1, 2, 3]は [[], [1], [2], [3], [1, 2], [2, 3], [1, 3], [1, 2, 3]]

になる。（ただし、順番はこの通りでなくても良い。）

3.9 代数的データ型の定義

リストのようなデータ型をプログラマがあらたに定義する方法も用意されている。このようなデー
タ型は複数の を持つことができる。このようなデータ型を （algebraic

datatype）という。
データ型の宣言の一般的な形式は次のとおりである。

data 型構成子名 型引数 1 型引数 2 . . . 型引数 k
= 構成子名 1 型 1,1 . . . 型 1,n1

| 構成子名 2 型 2,1 . . . 型 2,n2
| . . .
| 構成子名 m 型 m,1 . . . 型 m,nm

型構成子名・構成子名ともに使える文字は変数名の場合と同じだが、変数名とは逆に から始
まる必要がある。
代数的データ型はパラメータがない場合は、Cの列挙（enum）型と同じようなものである。次の例

では、
data Direction = Up | Down | Left | Right

Up, Down, Left, Rightの 4つがが Direction型を構成している。
一般的には代数的データ型の構成子はパラメータを持つ。次の例は二分木を表すデータ型を定義し

ている。
data Tree a = Branch (Tree a) a (Tree a) | Empty

このデータ型は Branchと Emptyの 2つの構成子を持つ。Emptyは引数を取らず、それだけで二分木
を構成する。Branchは 3つのパラメータを取る。1番目と 3番目は自分自身と同じ型の二分木であ
り、2番目は要素である。つまり、Branchは次のような型を持っている。

Branch ::

ここで、aは であり、ここには Integerや Stringなどの具体化な型がはいるこ
とができる。例えば Tree Integerは要素が Integer型であるような二分木の型である。Tree自体
は型ではなく型構成子 (type constructor)である。つまり、型パラメータを伴ってはじめて型になる。
具体的には次のような式がこの Tree型を持つ。

Empty :: Tree a
Branch Empty 1 Empty :: Tree Integer
Branch (Branch Empty "a" Empty) "b" (Branch Empty "c" Empty) :: Tree String
Branch (Branch Empty 1 Empty) 2 (Branch Empty 3 (Branch Empty 4 Empty))
:: Tree Integer

例えば最後の式の構造は、図で表すと次のようになる。

問 3.9.1 Tree型に対して、次のような関数を定義せよ。

size :: Tree a -> Integer -- 要素数
depth :: Tree a -> Integer -- 深さ
preorder :: Tree a -> [a] -- 前順走査
postorder :: Tree a -> [a] -- 後順走査
reflect :: Tree a -> Tree a -- 鏡像

3.10 Haskellの評価戦略

Haskellの評価戦略は基本的にラムダ計算のところで紹介した による評価方法である。

つまり正規形を持つ式の評価は必ず止まる。ただし、内部的には を用いる点がラムダ
計算の場合と異なる。
例えば、次のように定義された twiceという関数を考える。
twice x = x*x

最左戦略では twice (2+2)という式は次のように計算することになる。

twice (2 + 2) = (2 + 2) ∗ (2 + 2)

= 4 ∗ 4

= 16

つまり、twiceの引数である (2+2)は計算されないまま、まず twiceの定義にしたがって式が展開
される。そして、本当に必要になって (この場合は*の引数だから必要)はじめて 2+2が計算される。
この方式をナイーブに実行すると、2+2が二度計算されてしまう。
グラフ簡約では、このような計算をグラフの形で表して、2+2を一度しか計算しないようにして
いる。

twice

²²ÁÁ
ÁÁ
ÁÁ

+
ÂÂ?

?
ÄÄÄÄÄ

22

−→
*

¡¡ÁÁ
+

ÂÂ?
?

ÄÄÄÄ
22

−→
*

¡¡ÁÁ
4

−→
16

最左戦略は必要になるまで評価を遅らせるので （ lazy evaluation）とも言われる。プロ
グラミング言語で遅延評価を用いる場合は、このようにグラフ簡約と組み合わせて、同じ計算の繰り
返しを避けるようにするのが一般的である。
遅延評価の良いところは概念的に無限の大きさのデータ構造を扱えることである。例えば次のよう

な関数を考える。

from :: Integer -> [Integer]
from n = n : from (n+1)

-- takeは Preludeに定義済み
take :: Integer -> [a] -> [a]
take 0 _ = []
take _ [] = []
take n (x:xs) = x : take (n-1) xs

すると from 1は [1, 2, 3, . . . ]という無限リストだが、この無限リストを途中で用いている
take 3 (from 1)という式は
take 3 (from 1)→ take 3 (1:from (1+1))→ 1:(take 2 (from (1+1)))

→ 1:(take 2 ((1+1):from (1+1+1)))→ 1:(1+1):(take 1 (from (1+1+1)))→ · · ·
→ 1:(1+1):(1+1+1):(take 0 (from (1+1+1+1)))→ 1:(1+1):(1+1+1):[] (=[1, 2, 3])

のように有限時間で計算できる。
なお、fromで生成されるような等差数列については..を使った略記法がいくつか用意されている。

Prelude> [1..]
[1,2,3,4,5,6,7,8,9,10,11, . . .
Prelude> [2,4..]
[2,4,6,8,10,12,14,16,18,20,22, . . .
Prelude> [1..10]
[1,2,3,4,5,6,7,8,9,10]
Prelude> [1,4..20]
[1,4,7,10,13,16,19]

遅延評価を用いるといろいろと興味深いプログラミングが可能になる。参考文献 [6]は、遅延評価
を利用したプログラムの部品化の手法を詳しく説明している。

問 3.10.1 takeの反対に、リストの最初の n個の要素を取り除く関数 myDrop :: Integer -> [a]

-> [a]を定義せよ。

問 3.10.2 fibをフィボナッチ数列の無限リストとして定義せよ。

問 3.10.3要素に重複のない昇順に並んだ 2つのリストをマージして、やはり重複なしの昇順のリス
トを生成する関数 mergeを定義せよ。

問 3.10.4 2i · 3 j · 5k（ i, j, kは 0以上の整数）の形で表せる整数のみを重複なしに昇順に並べたリスト
hammingを定義せよ。

問 3.10.5 Haskell以外の言語で、無限リストをシミュレートする方法を考察せよ。またそれを用いて、
素数列を生成せよ。

3.11 リストの内包表記（List Comprehension）

Haskellは、リストの （List Comprehension）という糖衣構文（syntax sugar）を持つ。
これは数学で使われる集合の内包表記に似た記法である。

例
Prelude> [(x, y) | x <- [1, 2, 3, 4], y <- [5, 6, 7]]
[(1,5),(1,6),(1,7),(2,5),(2,6),(2,7),(3,5),(3,6),(3,7),(4,5),(4,6),(4,7)]
Prelude> [x*x | x <- [1..10], odd x]
[1,9,25,49,81]

（ただし [1..10]は [1,2,3,4,5,6,7,8,9,10]の略記法である。）
リストの内包記法は次のような形のものである。
[ 式 | 限定式, . . . , 限定式 ]

ここで限定式は、Bool型の式（ガード）か、次の形（生成式）:
変数（一般にはパターン） <- 式

のいずれかである。生成式の左辺の変数のスコープは、それより後の限定式である。生成式で変数に
右辺の式を評価して得られるリストの要素を順に代入し、ガードで真となるもののみ抽出して、すべ
ての組み合わせを列挙する。

問 3.11.1非負の整数 nを受け取り、0 ≤ x ≤ y ≤ nとなるすべての x, yの組を生成する関数 foo ::

Integer -> [(Integer, Integer)]を内包表記を用いて定義せよ。

問 3.11.2非負の整数 nを受け取り、0 < x < y < z≤ nの範囲で x2 + y2 = z2となるすべての x, y, zの
組を生成する関数 chokkaku :: Integer -> [(Integer, Integer, Integer)]を内包表記を用
いて定義せよ。

翻訳 リストの内包表記は次のような高階関数を用いて翻訳することができる。
-- 要素数 1のリストを返す
unit :: a -> [a]
unit a = a : []

bind :: [a] -> (a -> [b]) -> [b]
bind [] _ = []
bind (x:xs) f = append (f x) (bind xs f)

翻訳規則は次のとおりである。

[t] ⇒ unit t

[t | x <- u, P] ⇒ bind u (\ x -> [t | P])

[t | b, P] ⇒ if b then [t | P] else []

ただし、bは Bool値の式、Pは限定式のならびである。
内包表記を用いると、クイックソートは次のように簡潔に表される。
qsort [] = []
qsort (x:xs) = qsort [ y | y <- xs, y < x] ++ x : qsort [ y | y <- xs, y >= x]

問 3.11.3次の内包記法を上の翻訳規則を用いて、unit, bindを用いた形にせよ。

1. [(x, y) | x <- [1, 2, 3, 4], y <- [5, 6, 7]]

2. [x*x | x <- [1..10], odd x]

問 3.11.4 primesを素数列 [2, 3, 5, 7, 11, . . . ]の無限リストとして定義せよ。（内包表記を用
いても、用いなくても良い。）

考え方: “エラトステネス (Eratosthenes)のふるい” というアルゴリズムを実装する。このおなじみ
のアルゴリズムを言葉で表現すると次のようになる。

1. 2以上の自然数を並べる。

2. 先頭の数を取り除き、その倍数を同時にとり除く。

3. 2を繰り返す。

この時に先頭に現れた数を順番に並べたものが素数の列である。途中で無限リストの無限リストが現
われるが、問題ない。
このようにして、素数を無限リストとして表現することで、さまざまな “境界条件” に対応するこ

とができる。Cなどで実装しようとすると、1000までの素数というのは配列を用いて簡単に求める
ことができるが、最初の 100個の素数を求めるのは急に難しくなる。

3.12 8クイーンの問題

リストの内包表記を用いて、有名なパズルを解いてみることにする。
8クイーンの問題は 8個のクイーンを、お互いにとることができないように、チェス盤の上に置く

という問題である。可能な解はいくつか存在する。この可能な解の集まりをリストとして表現する。
ここでは無限リストは本質的には使用しないが、遅延評価は効率の点で決定的な役割を果たす。
クイーンの配置は、ここでは数のリストで表す。[4, 6, 1, 5, 2, 8, 3, 7]は次のような配置

を表す。

safe p nは、length p列までのクイーンの配置が pというリストで与えられた時、第 length p

+ 1列の第 n行にクイーンを置くことができるかどうかを示す関数である。
safe p n = all not [ check (i, j) (1+length p, n) | (i, j) <- zip [1..] p ]

check (i,j) (m,n) = j==n || (i+j==m+n) || (i-j==m-n)

ここで、allは
all :: (a -> Bool) -> [a] -> Bool
all p [] = True
all p (x:xs) = if p x then all p xs else False

と定義された標準ライブラリの関数である。[1..]は [1, 2, 3, . . . ]という無限リストである。

> safe [1, 3] 5
True
> safe [1, 3] 2
False

となる。

�

�

�

�

�

�
順に、最初の m列のすべての安全な配置を調べていく、そのために、そのようなすべての配置 (の
リスト)を返す queensという関数を定義する。

queens 0 = [[]]
queens m = [ append p [n] | p<-queens (m-1), n<-[1..8], safe p n ]

例えば

> queens 1
[[1], [2], [3], [4], [5], [6], [7], [8]]
> queens 2
[[1,3],[1,4],[1,5],[1,6],[1,7],[1,8],[2,4],[2,5],[2,6],[2,7],[2,8],
[3,1],[3,5],[3,6],[3,7],[3,8],[4,1],[4,2],[4,6],[4,7],[4,8],[5,1],
[5,2],[5,3],[5,7],[5,8],[6,1],[6,2],[6,3],[6,4],[6,8],[7,1],[7,2],
[7,3],[7,4],[7,5],[8,1],[8,2],[8,3],[8,4],[8,5],[8,6]]

となる。そうすると head (queens 8)で最初の解を求めることができる。

> head (queens 8)
[1,5,8,6,3,7,2,4]

遅延評価を用いているので、最初の解を求めるためには本当に必要な部分の簡約しか行なわない。つ
まり、上の [1, 5, 8, 6, 3, 7, 2, 4]を求めるのに、 queens 7の計算をすべて行なっているわ
けではなく、この最初の解を求めるのに必要なだけの部分の計算をしている。これは、queens 7を
完全に計算させると head (queens 8)よりも計算に時間がかかることからもわかる。ここでは、遅
延評価は Prologでいうところの後戻り (backtrack)と同じような効果を実現する。1つだけ解を求め
る計算とすべての解を求める計算を別々に記述する必要はなく、しかも、1つだけ解を求めるために
は、それに必要なだけの計算しか行なわない
ちなみに queens 8を計算させると、

> queens 8
[[1,5,8,6,3,7,2,4],[1,6,8,3,7,4,2,5],
(略)
,[8,3,1,6,2,5,7,4],[8,4,1,3,6,2,7,5]]

解はすべてで 92個あることがわかる。

